*===============================================================================
* Do-file creates "Table A4: Administrative Dataset: Violence and M-Paisa Use, Conditioning on Prior Exposure to Violence"
*===============================================================================
clear 
set more off

cd "~/Dropbox/VFD/REStat_Native_Files/Stata_Files"
local output "~/Dropbox/VFD/REStat_Native_Files/Stata_Files/Output"



use "monthpanel_analysis.dta", clear
keep if salary==1
gen trimmedobs=0
sum balance, d
replace trimmedobs =1 if (balance <=r(p99) & balance>=r(p1))
replace trimmedobs =. if balance ==.
bys identityid: egen trimmed_extusers = min(trimmedobs)
keep if trimmed_extusers==1


foreach v in violence df idf ied {
	bys identityid (ym): gen `v'_tot = sum(`v'_10km)
	gen d_`v'_tot=(`v'_tot>0)
}

foreach v in violence df idf ied {
	gen l_`v'_tot = l1.`v'_tot
	label var l_`v'_tot "Total Violent Events up to t-1"
}

estimates clear
	foreach y in balance numtxns numwithdrawal numdeposit numtransferdebit { 
		reghdfe `y' d_violence_10km l_violence_tot, absorb(ym identityid c.ym#district) cluster(district)
			est store `y'
			estadd ysumm, mean prefix(in_sample_)
			unique identityid if e(sample)
			estadd local nID = r(unique)
			estadd local samplestr "Salary Users"
			estadd local monthFE "YES"
			estadd local distFE "YES"
			estadd local individFE "YES"
	}
	estout balance numtxns numwithdrawal numdeposit numtransferdebit ///
	using "`output'/table_a4.tex", cells(b(fmt(%9.3f) star) se(par fmt(%9.3f))) ///
	starlevels(* .1 ** .05 *** .01) style(tex) keep(d_violence_10km l_violence_tot) mlabels(, dep) collabels(,none) ///
	stats(samplestr in_sample_mean nID N_clust N r2 monthFE individFE distFE, fmt(0 3 0 0 0 3 0 0 0) ///
	label("Sample" "Mean Dep Var" "\# Individuals" "\# Clusters" "\# Observations" "R-Squared" "Month-Year FE" "Individual FE" "District Trends")) ///
	label title("Administrative Dataset: Violence and M-Paisa Use") replace


